NRL Report 7333

A Computer Program for Determining the Polycrystalline Elastic Moduli from the Single-Crystal Elastic Constants

C. L. VOLD

Transformations and Kinetics Branch
Metallurgy Division

November 30, 1971

NAVAL RESEARCH LABORATORY
Washington, D.C.

UNCLASSIFIED

CONTENTS

ABSTRACT

A computer program has been written to evaluate the effective elastic moduli of a quasi-isotropic (random) polycrystalline aggregate, in terms of the single-crystal elastic constants. These moduli are calculated using the variational methods developed by Hashin and Shtrikman, and establish stringent limits on the upper and lower bounds for the effective elastic moduli. The program may be applied to crystals of cubic (all point groups), hexagonal (all point groups), trigonal (point groups $\bar{3}$m, 32, and 3m) and tetragonal (point groups 4/mmm, $\bar{4}$2m, 4mm, and 422) symmetries.

PROBLEM STATUS

This is a final report on one phase of the problem; work on other phases is continuing.

AUTHORIZATION

NRL Problem M01-23
Project RR 002-01-46-5430

Manuscript submitted July 22, 1971.

UNCLASSIFIED

A COMPUTER PROGRAM FOR DETERMINING THE POLYCRYSTALLINE ELASTIC MODULI FROM THE SINGLE-CRYSTAL ELASTIC CONSTANTS

INTRODUCTION

The problem of determining the effective elastic moduli of a quasi-isotropic (random) polycrystalline aggregate in terms of the single-crystal elastic constants has received a great deal of attention over the past eighty years. The first contribution was that of Voigt (1), who assumed that when a polycrystalline specimen is subjected to a gross uniform strain, the crystals would all adopt the same state of applied uniform *strain*. An analogous approach was then developed by Reuss (2), who assumed instead that the crystals would all adopt the same state of uniform *stress*. Hill (3) has shown that both models are only approximate, because in Voigt's model the forces between the grains would not be in equilibrium, and in Reuss' model the distorted gains would not fit together. Hill has also shown that the Voigt and Reuss moduli represent a set of upper and lower bounds, respectively, for the effective elastic modulus, and has suggested that the arithmetic or geometric mean of these bounds, now often referred to as the Hill average, might closely approximate the true value. Although no theoretical justification for the Hill average exists, this average appears empirically to be correct for a number of materials (4,5). It should be noted, however, that experimental measurements of moduli for certain materials may be highly variable, particularly when they are derived from static or low-frequency dynamic techniques. Both of these techniques are subject to serious errors arising from anelastic grain-boundary relaxation, viscous creep, and dislocation motion — effects which become increasingly important at elevated temperatures (6). Also, slight crystallographic texturing in the specimens can lead to high variable elastic moduli (7).

Various other attempts to refine and improve the theory can be cited. These include the efforts of Bruggeman (8), Boas (9), Laurent and Eudier (10), Hershey (11), Kröner (12), Kajamaa (13), and Huber and Schmid (14). The latter authors' procedure for calculating the Young's modulus is equivalent to assuming that, in any grain, the stress is a pure tension parallel to some fixed direction and of an amount sufficient to produce a unit extension in that direction. A Young's modulus is defined for each direction in the crystal and is averaged over all possible directions. In the Reuss theory, on the other hand, it is the inverse of the modulus that is averaged. Therefore, the Huber-Schmid Young's modulus must exceed the Reuss estimate of the Young's modulus. However, it has not been proved that the Huber-Schmid modulus is less than the true Young's modulus, or even less than the Voigt estimate of the Young's modulus.

More recently, Hashin and Shtrikman (15,16) have developed variational principles for anisotropic and nonhomogeneous elastic media. These principles establish a rigorous method for obtaining more stringent limits on the upper and lower bounds for the effective elastic moduli than provided previously. Hashin and Shtrikman avoid any assumptions about the grain shapes in the polycrystal and merely require a random distribution of crystallographic orientations. Although the method was presented as a general one, the theoretical results of Hashin and Shtrikman were applied only to crystals of cubic symmetry. In the case of cubic symmetry the Voigt and Reuss bounds of the bulk modulus coincide; thus, only the bounds for the shear moduli had to be considered. The results of Hashin and Shtrikman yielded considerably improved bounds for the effective shear moduli of cubic materials. The Hashin-Shtrikman method has recently been extended by Peselnick and Meister (17,18) to cases of lower crystallographic symmetry, where the effective bulk modulus, as well as the effective shear (or Young's) modulus, must be calculated from the single-crystal elastic constants.

METHOD

This report will present a computer program, written in 3600 Fortran, used to evaluate the Hashin-Shtrikman variational bounds of the elastic moduli for crystals of cubic, hexagonal, trigonal, and tetragonal symmetries. For details concerning the theory of the procedure, the reader is referred to the original work of Hashin and Shtrikman (15,16), and of Peselnick and Meister (17,18). In short, the greatest lower bound and the least upper bound of the elastic moduli are expressed in terms of the variables G_i,K_i. The procedure employed to find the values of $G_i,K_i = G_1,K_1$, which produce the greatest lower bound of the modulus, and $G_i,K_i = G_2,K_2$, which produce the least upper bound of the modulus, must be consistent with the positive (or negative) definite requirements of a fourth rank tensor R_{ijkl}, defined by

$$R_{ijkl} = C_{ijkl} - C^o{}_{ijkl} \tag{1}$$

where C_{ijkl} and $C^o{}_{ijkl}$ are the anisotropic and the average isotropic elastic moduli tensors, respectively. The equations necessary for arriving at the appropriate values of G_i,K_i, and subsequently the variational bounds of the bulk and shear moduli, are summarized in the following sections. The variational bounds for the Young's modulus E_i^* may then be computed from the isotropic linear elastic relation

$$E_i^* = \frac{9K_i^*G_i^*}{3K_i^* + G_i^*}. \tag{2}$$

The following definitions apply:

K_V = Voigt estimate of the bulk modulus

K_R = Reuss estimate of the bulk modulus

G_V = Voigt estimate of the shear modulus

G_R = Reuss estimate of the shear modulus.

CUBIC SYMMETRY (All Point Groups)

For crystals exhibiting cubic symmetry, the independent C_{ij}'s are C_{11}, C_{12}, and C_{44}. Here,

$$K_V = K_R = (C_{11} + 2C_{12})/3, \tag{3}$$

$$G_V = (C_{11} - C_{12} + 3C_{44})/5, \tag{4}$$

and

$$G_R = \frac{5C_{44}(C_{11} - C_{12})}{4C_{44} + 3(C_{11} - C_{12})}. \tag{5}$$

The variational bounds for the shear moduli are given by

$$G_i^* = G_i + \frac{(B_2)_i}{1 + 2\beta_i(B_2)_i}, \tag{6}$$

where

$$\beta_i = \frac{-3(K_i + 2G_i)}{5G_i(3K_i + 4G_i)} \tag{7}$$

and

$$(B_2)_i = \frac{1}{5}\left(\frac{C_{11} - C_{12} - 2G_i}{1 - \beta_i(C_{11} - C_{12} - 2G_i)}\right) + \frac{3}{5}\left(\frac{C_{44} - G_i}{1 - 2\beta_i(C_{44} - G_i)}\right). \tag{8}$$

For the greatest lower bound, $G_i, K_i = G_1, K_i$ where

$$G_1 = C_{44}, \text{ or } G_1 = (C_{11} - C_{12})/2, \tag{9}$$

whichever is smaller, and

$$K_i = (C_{11} + 2C_{12})/3. \tag{10}$$

For the least upper bound, $G_i, K_i = G_2, K_i$ where

$$G_2 = C_{44}, \text{ or } G_2 = (C_{11} - C_{12})/2, \tag{11}$$

whichever is larger, and K_i is given by Eq. (10).

HEXAGONAL SYMMETRY (All Point Groups)

For crystals exhibiting hexagonal symmetry, the independent C_{ij}'s are C_{11}, C_{12}, C_{13}, C_{33}, and C_{44}, with

$$C_{66} = (C_{11} - C_{12})/2.$$

Here,

$$K_V = (2(C_{11} + C_{12}) + C_{33} + 4C_{13})/9 \tag{12}$$

and

$$K_R = C^2/M \tag{13}$$

where

$$C^2 = (C_{11} + C_{12})C_{33} - 2C_{13}^2 \tag{14}$$

and

$$M = C_{11} + C_{12} + 2C_{33} - 4C_{13}. \tag{15}$$

Also,

$$G_V = (12C_{66} + 12C_{44} + M)/30 \tag{16}$$

and

$$G_R = \frac{5}{2}\left(\frac{C_{44}C_{66}C^2}{(C_{44} + C_{66})C^2 + 3K_V C_{44}C_{66}}\right). \tag{17}$$

The variational bounds for the bulk moduli are given by

$$K_i^* = K_i + \frac{K_V - K_i}{1 - \beta_i(C_{11} + C_{12} + C_{33} - 3K_V - 2G_i)}, \tag{18}$$

and the variational bounds for the shear moduli are given by

$$G_i^* = G_i + \frac{(B_2)_i}{1 + 2\beta_i(B_2)_i}. \tag{6}$$

Here,

$$\beta_i = \frac{-3(K_i + 2G_i)}{5G_i(3K_i + 4G_i)} \tag{19}$$

and

$$30(B_2)_i = \frac{M - 6G_i}{1 - \beta_i(C_{11} + C_{12} + C_{33} - 3K_i - 2G_i) - 9\gamma_i(K_V - K_i)} + \frac{12(C_{66} - G_i)}{1 - 2\beta_i(C_{66} - G_i)} + \frac{12(C_{44} - G_i)}{1 - 2\beta_i(C_{44} - G_i)} \tag{20}$$

where

$$\gamma_i = (\alpha_i - 3\beta_i)/9, \tag{21}$$

and

$$\alpha_i = \frac{-3}{3K_i + 4G_i}. \tag{22}$$

For the greatest lower bound, $G_i, K_i = G_1, K_1$ where

$$G_1 = C_{44} \text{ or } G_1 = C_{66}, \tag{23}$$

whichever is smaller, and

$$K_1 = \frac{C^2 - 6G_1 K_V}{M - 6G_1}. \tag{24}$$

For the least upper bound, $G_i, K_i = G_2, K_2$ where

$$G_2 = C_{44}, \text{ or } G_2 = C_{66}, \tag{25}$$

whichever is larger, and

$$K_2 = \frac{C^2 - 6G_2 K_V}{M - 6G_2}. \tag{26}$$

TRIGONAL SYMMETRY (Point Groups $\bar{3}$m, 32, 3m)

For crystals exhibiting trigonal symmetry (Laue class $\bar{3}$m), the independent C_{ij}'s are C_{11}, C_{12}, C_{13}, C_{14}, C_{33}, and C_{44}, with

$$C_{66} = (C_{11} - C_{12})/2.$$

Here,

$$K_V = (2(C_{11} + C_{12}) + C_{33} + 4C_{13})/9 \tag{12}$$

and

$$K_R = C^2/M \tag{13}$$

where

$$C^2 = (C_{11} + C_{12})C_{33} - 2C_{13}^2 \tag{14}$$

and

$$M = C_{11} + C_{12} + 2C_{33} - 4C_{13}. \tag{15}$$

Also,

$$G_V = (12C_{66} + 12C_{44} + M)/30 \tag{16}$$

and

$$G_R = \frac{5C^2(C_{44}C_{66} - C_{14}^2)}{[6K_V(C_{44}C_{66} - C_{14}^2) + 2C^2(C_{44} + C_{66})]}. \tag{27}$$

The variational bounds for the bulk moduli are given by

$$K_i^* = K_i + \frac{K_V - K_i}{1 - \beta_i(C_{11} + C_{12} + C_{33} - 3K_V - 2G_i)}, \tag{18}$$

and the variational bounds for the shear moduli are given by

$$G_i^* = G_i + \frac{(B_2)_i}{1 + 2\beta_i(B_2)_i}. \tag{6}$$

In these expressions

$$30(B_2)_i = \frac{M - 6G_i}{1 - \beta_i(C_{11} + C_{12} + C_{33} - 3K_i - 2G_i) - 9\gamma_i(K_V - K_i)} + \frac{12(C_{44} + C_{66} - 2G_i)}{1 - \beta_i(C_{44} + C_{66} - 2G_i)}, \tag{28}$$

and β_i, α_i, and γ_i are given by Eqs. (7), (22), and (21), respectively.

For the greatest lower bound, $G_i, K_i = G_1, K_1$ where

$$G_1 = 1/2(C_{44} + C_{66}) - (1/4(C_{44} - C_{66})^2 + C_{14}^2)^{1/2} \tag{29}$$

and

$$K_1 = \frac{C^2 - 6G_1K_V}{M - 6G_1}. \tag{24}$$

For the least upper bound, $G_i, K_i = G_2, K_2$ where

$$G_2 = 1/2(C_{44} + C_{66}) + (1/4(C_{44} - C_{66})^2 + C_{14}^2)^{1/2} \tag{30}$$

UNCLASSIFIED

and

$$K_2 = \frac{C^2 - 6G_2K_V}{M - 6G_2}. \tag{26}$$

TETRAGONAL SYMMETRY (Point Groups 4/mmm, $\bar{4}$2m, 4mm, and 422)

For crystals of tetragonal symmetry (Laue class 4/mmm), the independent C_{ij}'s are C_{11}, C_{12}, C_{13}, C_{33}, C_{44}, and C_{66}. Here,

$$K_V = (2(C_{11} + C_{12}) + C_{33} + 4C_{13})/9 \tag{12}$$

and

$$K_R = C^2/M \tag{13}$$

where

$$C^2 = (C_{11} + C_{12})C_{33} - 2C_{13}^2 \tag{14}$$

and

$$M = C_{11} + C_{12} + 2C_{33} - 4C_{13}. \tag{15}$$

Also,

$$G_V = (M + 12C_{44} + 6C_{66} + 3C_{11} - 3C_{12})/30 \tag{31}$$

and

$$G_R = 15\left(\frac{(18K_V + C_{33})}{C^2} + \frac{6}{C_{11} - C_{12}} + \frac{6}{C_{44}} + \frac{3}{C_{66}}\right)^{-1}. \tag{32}$$

The variational bounds for the bulk and shear moduli are given by the expressions

$$K_i^* = K_i + \frac{K_V - K_i}{1 - \beta_i(C_{11} + C_{12} + C_{33} - 3K_V - 2G_i)} \tag{18}$$

and

$$G_i^* = G_i + \frac{(B_2)_i}{1 + 2\beta_i(B_2)_i}, \tag{6}$$

respectively, where β_i is defined by Eq. (19), and

$$15(B_2)_i = \frac{M - 6G_i + (C_{11} - C_{12} - 2G_i)\,[3 - 2\beta_i\theta_i - 27\gamma_i(K_V - K_i)]}{2[1 - \beta_i(C_{11} - C_{12} - 2G_i)]\,[1 - \beta_i\xi_i - 9\gamma_i(K_V - K_i)]}$$

$$+ \frac{6(C_{44} - G_i)}{1 - 2\beta_i(C_{44} - G_i)} + \frac{3(C_{66} - G_i)}{1 - 2\beta_i(C_{66} - G_i)}. \qquad (33)$$

In Eq. (33)

$$M = C_{11} + C_{12} + 2C_{33} - 4C_{13} \qquad (15)$$

$$\theta_i = 9K_V - (9/2)K_i - 6G_i + (3/2)C_{33} - 6C_{13}, \qquad (34)$$

$$\xi_i = C_{11} + C_{12} + C_{33} - 3K_i - 2G_i, \qquad (35)$$

$$\gamma_i = (\alpha_i - 3\beta_i)/9 \qquad (21)$$

and

$$\alpha_i = \frac{-3}{3K_i + 4G_i}. \qquad (22)$$

For the greatest lower bound, $G_i,K_i = G_1,K_1$ where

$$G_1 = C_{66}, \text{ or } G_1 = C_{44}, \text{ or } G_1 = (C_{11} - C_{12})/2, \qquad (36)$$

whichever is smaller, and

$$K_1 = \frac{C^2 - 6G_1K_V}{M - 6G_1}. \qquad (24)$$

For the least upper bound, $G_i,K_i = G_2,K_2$ where

$$G_2 = C_{66}, \text{ or } G_2 = C_{44}, \text{ or } G_2 = (C_{11} - C_{12})/2, \qquad (37)$$

whichever is larger, and

$$K_2 = \frac{C^2 - 6G_2K_V}{M - 6G_2}. \qquad (26)$$

REFERENCES

1. Voigt, W., Ann. Phys. (Leipzig), 38:573 (1889); "Lehrbuch der Kristallphysik," Leipzig: Teubner, 1910 (reprinted 1928).

2. Reuss, A., Z. Angew. Math. Mech., 9:49 (1929).

3. Hill, R., Proc. Phys. Soc. (London), A65:349 (1952).

4. Chung, D.H., and Buessem, W.R., J. Appl. Phys., 38:2535 (1967); 39:2777 (1968).

5. Hearmon, R.F.S., "An Introduction to Applied Anisotropic Elasticity," p. 44, London: Oxford University Press, 1961.

6. Gjostein, N.A., Acta Met., 8:263 (1960).

7. Hearmon, R.F.S., Advan. Phys., 5:323 (1956).

8. Bruggeman, D.A.G., Z. Phys., 92:561 (1934).

9. Boas, W., "An Introduction to the Physics of Metals and Alloys," Melbourne: University Press; New York: Wiley, 1947.

10. Laurent, P., and Eudier, M., Rev. Métall., 47:582 (1950).

11. Hershey, A.V., J. Appl. Mech., 21:236 (1954).

12. Kröner, E., Z. Phys., 151:504 (1958).

13. Kajamaa, J., Phil. Mag., 22:423 (1970).

14. Huber, A., and Schmid. E., Helv. Phys. Acta., 7:620 (1934).

15. Hashin, Z., and Shtrikman, S., J. Mech. Phys. Solids, 10:335 (1962).

16. Hashin, Z., and Shtrikman, S., J. Mech. Phys. Solids, 10:343 (1962).

17. Peselnick, L., and Meister, R., J. Appl. Phys., 36:2879 (1965).

18. Meister, R., and Peselnick, L., J. Appl. Phys., 37:4121 (1966).

APPENDIX A

"PROGRAM BOUND"

DATA CARD FORMAT

Field	Identifier	Format	
1	Q	A8	Alphanumeric Identification Characters
2	N	I2	N = 1 for Cubic System N = 2 for Hexagonal System N = 3 for Trigonal System N = 4 for Tetragonal System
3	C_{11}	F10.4	
4	C_{12}	F10.4	
5	C_{13}	F10.4	Single-Crystal Elastic Stiffness Constants
6	C_{14}	F10.4	
7	C_{33}	F10.4	
8	C_{44}	F10.4	
9	C_{66}	F10.4	

NOTE: All elastic stiffness constants for any given system must be entered, both independent and dependent. Unused fields may be left blank.

```
      PROGRAM BOUND
    6 FORMAT(A8,I2,7F10.4)
  129 FORMAT(3X,A8)
  130 FORMAT(3X,12HCUBIC SYSTEM/3X,10HINPUT DATA,5X,5HC11 =,F9.4,3X,
     15HC12 =,F9.4,3X,5HC44 =,F9.4/)
  140 FORMAT (3X,12HBULK MODULUS,32X,F10.4/
     13X,36HVARIATIONAL BOUNDS OF SHEAR MODULI   ,8X,2F10.4,
     220X,9HAVERAGE =,F10.4/
     33X,36HVARIATIONAL BOUNDS OF YOUNGS MODULI ,8X,2F10.4,
     420X,9HAVERAGE =,F10.4)
  150 FORMAT(3X,39HVOIGT AND REUSS BOUNDS OF SHEAR MODULI ,5X,2F10.4,
     120X,9HAVERAGE =,F10.4)
  160 FORMAT(3X,39HVOIGT AND REUSS BOUNDS OF YOUNGS MODULI,5X,2F10.4,
     120X,9HAVERAGE =,F10.4////)
  240 FORMAT(3X,16HHEXAGONAL SYSTEM/3X,10HINPUT DATA,5X,5HC11 =,F9.4,3X,
     15HC12 =,F9.4,3X,5HC13 =,F9.4,3X,5HC33 =,F9.4,3X,5HC44 =,F9.4,3X,
     25HC66 =,F9.4/)
  250 FORMAT(3X,36HVARIATIONAL BOUNDS OF BULK MODULI    ,8X,2F10.4,
     120X,9HAVERAGE =,F10.4/
     23X,36HVARIATIONAL BOUNDS OF SHEAR MODULI   ,8X,2F10.4,
     320X,9HAVERAGE =,F10.4/
     43X,36HVARIATIONAL BOUNDS OF YOUNGS MODULI ,8X,2F10.4,
     520X,9HAVERAGE =,F10.4)
  260 FORMAT(3X,39HVOIGT AND REUSS BOUNDS OF BULK MODULI   ,5X,2F10.4,
     120X,9HAVERAGE =,F10.4/
     23X,39HVOIGT AND REUSS BOUNDS OF SHEAR MODULI ,5X,2F10.4,
     320X,9HAVERAGE =,F10.4/
     43X,39HVOIGT AND REUSS BOUNDS OF YOUNGS MODULI,5X,2F10.4,
     520X,9HAVERAGE =,F10.4////)
  310 FORMAT(3X,15HTRIGONAL SYSTEM/3X,10HINPUT DATA,5X,5HC11 =,F9.4,3X,
     15HC12 =,F9.4,3X,5HC13 =,F9.4,3X,5HC14 =,F9.4,3X,5HC33 =,F9.4,3X,
     25HC44 =,F9.4,3X,5HC66 =,F9.4/)
  480 FORMAT(3X,17HTETRAGONAL SYSTEM/3X,10HINPUT DATA,5X,5HC11 =,F9.4,3X
     1,5HC12 =,F9.4,3X,5HC13 =,F9.4,3X,5HC33 =,F9.4,3X,5HC44 =,F9.4,3X,
     25HC66 =,F9.4,/)
    5 READ 6, Q,N,C11,C12,C13,C14,C33,C44,C66
      IF (EOF,60)500,10
   10 IF (N.EQ.1)100,20
   20 IF (N.EQ.2)200,30
   30 IF (N.EQ.3)300,40
   40 GO TO 400
  100 CONTINUE
C     CRYSTAL SYSTEM IS CUBIC (ALL CLASSES)
      UV=(C11+2*C12)/3.
      UR=(C11+2*C12)/3.
      GV=(C11-C12+3*C44)/5.
      U1=(C11+2*C12)/3.
      U2=(C11+2*C12)/3.
      GR=( 5.*C44*(C11-C12))/(4.*C44+3.*C11-3.*C12)
      IF(C44.LE.(C11-C12)/2.)110,120
  110 G1=C44
      G2=(C11-C12)/2.
      GO TO 125
  120 G1=(C11-C12)/2.
      G2=C44
  125 BETA1=(-3.*(U1+2.*G1))/(5.*G1*(3.*U1+4.*G1))
      BETA2=(-3.*(U2+2.*G2))/(5.*G2*(3.*U2+4.*G2))
      B21=((C11-C12-2.*G1)*(1./5.))/(1.-(BETA1*(C11-C12-2.*G1)))
     1+((3./5.)*(C44-G1))/(1.-(2.*BETA1*(C44-G1)))
      B22=((C11-C12-2.*G2)*(1./5.))/(1.-(BETA2*(C11-C12-2.*G2)))
```

```
     1+((3./5.)*(C44-G2))/(1.-(2.*BETA2*(C44-G2)))
      GSTAR1=G1+(B21/(1.+(2.*BETA1*B21)))
      GSTAR2=G2+(B22/(1.+(2.*BETA2*B22)))
      YSTAR1=(9.*U1*GSTAR1)/(3.*U1+GSTAR1)
      YSTAR2=(9.*U2*GSTAR2)/(3.*U2+GSTAR2)
      YV=(9.*UV*GV)/(3.*UV+GV)
      YR=(9.*UR*GR)/(3.*UR+GR)
      YSTARAVE=(YSTAR1+YSTAR2)/2.
      GSTARAVE=(GSTAR1+GSTAR2)/2.
      YHILL=(YV+YR)/2.
      GHILL=(GV+GR)/2.
      PRINT 129,Q
      PRINT 130,C11,C12,C44
      PRINT 140,U1,GSTAR2,GSTAR1,GSTARAVE,YSTAR2,YSTAR1,YSTARAVE
      PRINT 150,GV,GR,GHILL
      PRINT 160,YV,YR,YHILL
      GO TO 5
  200 CONTINUE
C     CRYSTAL SYSTEM IS HEXAGONAL (ALL CLASSES)
      UV=(2.*(C11+C12)+C33+4.*C13)/9.
      Z=C11+C12+2.*C33-4.*C13
      CSQR=(C33*(C11+C12)-2.*C13*C13)
      UR=CSQR/Z
      GV=(12.*C66+12.*C44+Z)/30.
      GR=((5./2.)*C44*C66*CSQR)/((C44+C66)*CSQR+3.*UV*C44*C66)
      IF(C44.LE.C66)210,220
  210 G1=C44
      G2=C66
      GO TO 230
  220 G1=C66
      G2=C44
  230 U1=(CSQR-(6.*G1*UV))/(Z-6.*G1)
      U2=(CSQR-(6.*G2*UV))/(Z-6.*G2)
      BETA1=(-3.*(U1+2.*G1))/(5.*G1*(3.*U1+4.*G1))
      BETA2=(-3.*(U2+2.*G2))/(5.*G2*(3.*U2+4.*G2))
      BSTAR1=U1+((UV-U1)/(1.-BETA1*(C11+C12+C33-3.*UV-2.*G1)))
      BSTAR2=U2+((UV-U2)/(1.-BETA2*(C11+C12+C33-3.*UV-2.*G2)))
      ALPHA1=(-3.)/(3.*U1+4.*G1)
      ALPHA2=(-3.)/(3.*U2+4.*G2)
      GAMMA1=(ALPHA1-3.*BETA1)/9.
      GAMMA2=(ALPHA2-3.*BETA2)/9.
      B21=((Z-6.*G1)*(1./30.))/(1.-BETA1*(C11+C12+C33-3.*U1-2.*G1)
     1-9.*GAMMA1*(UV-U1))+((12.*(C66-G1))/30.)/(1.-2.*BETA1*(C66-G1))
     2+((12.*(C44-G1))/30.)/(1.-2.*BETA1*(C44-G1))
      B22=((Z-6.*G2)*(1./30.))/(1.-BETA2*(C11+C12+C33-3.*U2-2.*G2)
     1-9.*GAMMA2*(UV-U2))+((12.*(C66-G2))/30.)/(1.-2.*BETA2*(C66-G2))
     2+((12.*(C44-G2))/30.)/(1.-2.*BETA2*(C44-G1))
      GSTAR1=G1+B21/(1.+2.*BETA1*B21)
      GSTAR2=G2+B22/(1.+2.*BETA2*B22)
      YSTAR1=(9.*BSTAR1*GSTAR1)/(3.*BSTAR1+GSTAR1)
      YSTAR2=(9.*BSTAR2*GSTAR2)/(3.*BSTAR2+GSTAR2)
      YV=(9.*UV*GV)/(3.*UV+GV)
      YR=(9.*UR*GR)/(3.*UR+GR)
      YSTARAVE=(YSTAR1+YSTAR2)/2.
      GSTARAVE=(GSTAR1+GSTAR2)/2.
      YHILL=(YV+YR)/2.
      GHILL=(GV+GR)/2.
      BSTARAVE=(BSTAR1+BSTAR2)/2.
      BHILL=(UV+UR)/2.
      PRINT 129,Q
      PRINT 240, C11,C12,C13,C33,C44,C66
```

```
      PRINT 250,BSTAR2,BSTAR1,BSTARAVE,GSTAR2,GSTAR1,GSTARAVE,
     1YSTAR2,YSTAR1,YSTARAVE
      PRINT 260,UV,UR,BHILL,GV,GR,GHILL,YV,YR,YHILL
      GO TO 5
  300 CONTINUE
C     CRYSTAL SYSTEM IS TRIGONAL (CLASSES 3/M, 3BAR M, 32)
      UV=(2.*(C11+C12)+C33+4.*C13)/9.
      CSQR=(C33*(C11+C12)-2.*C13*C13)
      Z=C11+C12+2.*C33-4.*C13
      UR=CSQR/Z
      GV=(Z+12.*(C44+C66))/30.
      GR=(5.*CSQR*(C44*C66-C14*C14))/((6.*UV*(C44*C66-C14*C14))+(2.*
     1CSQR*(C44+C66)))
      G1=0.5*(C44+C66)-SQRT(0.25*((C44-C66)**2)+C14**2)
      G2=0.5*(C44+C66)+SQRT(0.25*((C44-C66)**2)+C14**2)
      U1=(CSQR-6.*G1*UV)/(Z-6.*G1)
      U2=(CSQR-6.*G2*UV)/(Z-6.*G2)
      BETA1=(-3.*(U1+2.*G1))/(5.*G1*(3.*U1+4.*G1))
      BETA2=(-3.*(U2+2.*G2))/(5.*G2*(3.*U2+4.*G2))
      ALPHA1=-3./(3.*U1+4.*G1)
      ALPHA2=-3./(3.*U2+4.*G2)
      GAMMA1=(ALPHA1-3.*BETA1)/9.
      GAMMA2=(ALPHA2-3.*BETA2)/9.
      BSTAR1=U1+((UV-U1)/(1.-BETA1*(C11+C12+C33-3.*UV-2.*G1)))
      BSTAR2=U2+((UV-U2)/(1.-BETA2*(C11+C12+C33-3.*UV-2.*G2)))
      B21=((Z-6*G1)*(1./30.))/(1.-BETA1*(C11+C12+C33-3.*U1-2.*G1)-9.*
     1GAMMA1*(UV-U1))+(12.*(C44+C66-2.*G1))/(30.*
     2(1.-2.*BETA1*(C44+C66-2.*G1)))
      B22=((Z-6*G2)*(1./30.))/(1.-BETA2*(C11+C12+C33-3.*U2-2.*G2)-9.*
     1GAMMA2*(UV-U2))+(12.*(C44+C66-2.*G2))/(30.*
     2(1.-2.*BETA2*(C44+C66-2.*G2)))
      GSTAR1=G1+B21/(1.+2.*BETA1*B21)
      GSTAR2=G2+B22/(1.+2.*BETA2*B22)
      YSTAR1=(9.*BSTAR1*GSTAR1)/(3.*BSTAR1+GSTAR1)
      YSTAR2=(9.*BSTAR2*GSTAR2)/(3.*BSTAR2+GSTAR2)
      YV=(9.*UV*GV)/(3.*UV+GV)
      YR=(9.*UR*GR)/(3.*UR+GR)
      YSTARAVE=(YSTAR1+YSTAR2)/2.
      GSTARAVE=(GSTAR1+GSTAR2)/2.
      BSTARAVE=(BSTAR1+BSTAR2)/2.
      YHILL=(YV+YR)/2.
      GHILL=(GV+GR)/2.
      BHILL=(UV+UR)/2.
      PRINT 129,Q
      PRINT 310,C11,C12,C13,C14,C33,C44,C66
      PRINT 250,BSTAR2,BSTAR1,BSTARAVE,GSTAR2,GSTAR1,GSTARAVE,
     1YSTAR2,YSTAR1,YSTARAVE
      PRINT 260,UV,UR,BHILL,GV,GR,GHILL,YV,YR,YHILL
      GO TO 5
  400 CONTINUE
C     CRYSTAL SYSTEM IS TETRAGONAL (CLASSES 4/MMM, 4BAR2M, 4MMM, 422)
      UV=(2./9.)*(C11+C12)+(C33/9)+(4./9.)*C13
      CSQR=(C11+C12)*C33-2*(C13**2)
      Z=C11+C12+2*C33-4*C13
      UR=CSQR/Z
      GR=15./((18.*UV+C33)/CSQR+6./(C11-C12)+6./C44+3./C66)
      GV=(Z+12.*C44+6.*C66+3.*C11-3.*C12)/30.
      IF(C44.LE.C66)410,435
  410 IF(C44.LE.(C11-C12)/2)415,430
  415 G1=C44
```

```
      IF(C66.GE.(C11-C12)/2)420,425
 420  G2=C66
      GO TO 460
 425  G2=(C11-C12)/2
      GO TO 460
 430  G1=(C11-C12)/2
      G2=C66
      GO TO 460
 435  IF(C66.LE.(C11-C12)/2)440,455
 440  G1=C66
      IF(C44.GE.(C11-C12)/2)445,450
 445  G2=C44
      GO TO 460
 450  G2=(C11-C12)/2
      GO TO 460
 455  G1=(C11-C12)/2
      G2=C44
      GO TO 460
 460  U1=(CSQR-6*UV*G1)/(Z-6*G1)
      U2=(CSQR-6*UV*G2)/(Z-6*G2)
      BETA1=-(3*U1+6*G1)/(15.*G1*U1+20.*G1*G1)
      BETA2=-(3*U2+6*G2)/(15.*G2*U2+20.*G2*G2)
      BSTAR1=U1+(UV-U1)/(1-BETA1*(C11+C12+C33-3*UV-2*G1))
      BSTAR2=U2+(UV-U2)/(1-BETA2*(C11+C12+C33-3*UV-2*G2))
      ALPHA1=-3/(3*U1+4*G1)
      ALPHA2=-3/(3*U2+4*G2)
      GAMMA1=(ALPHA1-3*BETA1)/9
      GAMMA2=(ALPHA2-3*BETA2)/9
      XI1 =C11+C12+C33-3*U1-2*G1
      XI2 =C11+C12+C33-3*U2-2*G2
      THETA1=9*UV-(9./2.)*U1-6*G1+(3./2.)*C33-6*C13
      THETA2=9*UV-(9./2.)*U2-6*G2+(3./2.)*C33-6*C13
      B21=(Z-6*G1+(C11-C12-2*G1)*(3-2*BETA1*THETA1-27*GAMMA1*(UV-U1)))/
     1(30*(1-BETA1*(C11-C12-2*G1))*(1-BETA1*XI1 -9*GAMMA1*(UV-U1)))
     2+(6*(C44-G1))/(15*(1-2*BETA1*(C44-G1)))
     3+(3*(C66-G1))/(15*(1-2*BETA1*(C66-G1)))
      B22=(Z-6*G2+(C11-C12-2*G2)*(3-2*BETA2*THETA2-27*GAMMA2*(UV-U2)))/
     1(30*(1-BETA2*(C11-C12-2*G2))*(1-BETA2*XI2 -9*GAMMA2*(UV-U2)))
     2+(6*(C44-G2))/(15*(1-2*BETA2*(C44-G2)))
     3+(3*(C66-G2))/(15*(1-2*BETA2*(C66-G2)))
      GSTAR1=G1+(B21/(1+2*BETA1*B21))
      GSTAR2=G2+(B22/(1+2*BETA2*B22))
      YSTAR1=(9*BSTAR1*GSTAR1)/(3*BSTAR1+GSTAR1)
      YSTAR2=(9*BSTAR2*GSTAR2)/(3*BSTAR2+GSTAR2)
      YV=(9.*UV*GV)/(3.*UV+GV)
      YR=(9.*UR*GR)/(3.*UR+GR)
      YSTARAVE=(YSTAR1+YSTAR2)/2.
      GSTARAVE=(GSTAR1+GSTAR2)/2.
      BSTARAVE=(BSTAR1+BSTAR2)/2.
      YHILL=(YV+YR)/2.
      GHILL=(GV+GR)/2.
      BHILL=(UV+UR)/2.
      PRINT 129,Q
      PRINT 480,C11,C12,C13,C33,C44,C66
      PRINT 250,BSTAR2,BSTAR1,BSTARAVE,GSTAR2,GSTAR1,GSTAVE,
     1YSTAR2,YSTAR1,YSTARAVE
      PRINT 260,UV,UR,BHILL,GV,GR,GHILL,YV,YR,YHILL
      GO TO 5
 500  STOP
      END
```

APPENDIX B

REPRESENTATIVE OUTPUT

IRON
CUBIC SYSTEM
INPUT DATA C11 = 23.7000 C12 = 14.1000 C44 = 11.6000

BULK MODULUS	17.3000		
VARIATIONAL BOUNDS OF SHEAR MODULI	8.3074	8.0451	AVERAGE = 8.1763
VARIATIONAL BOUNDS OF YOUNGS MODULI	21.4835	20.8962	AVERAGE = 21.1899
VOIGT AND REUSS BOUNDS OF SHEAR MODULI	8.8800	7.4043	AVERAGE = 8.1421
VOIGT AND REUSS BOUNDS OF YOUNGS MODULI	22.7479	19.4395	AVERAGE = 21.0937

COBALT
HEXAGONAL SYSTEM
INPUT DATA C11 = 30.7000 C12 = 16.5000 C13 = 10.3000 C33 = 35.8100 C44 = 7.5300 C66 = 7.1000

VARIATIONAL BOUNDS OF BULK MODULI	19.0435	19.0435	AVERAGE = 19.0435
VARIATIONAL BOUNDS OF SHEAR MODULI	8.1879	8.1828	AVERAGE = 8.1854
VARIATIONAL BOUNDS OF YOUNGS MODULI	21.4846	21.4729	AVERAGE = 21.4787
VOIGT AND REUSS BOUNDS OF BULK MODULI	19.0456	19.0422	AVERAGE = 19.0439
VOIGT AND REUSS BOUNDS OF SHEAR MODULI	8.4393	8.0050	AVERAGE = 8.2222
VOIGT AND REUSS BOUNDS OF YOUNGS MODULI	22.0597	21.0635	AVERAGE = 21.5616

BISMUTH
TRIGONAL SYSTEM
INPUT DATA C11 = 56.1000 C12 = 23.9000 C13 = 24.2000 C14 = 6.0000 C33 = 34.9000 C44 = 9.0000 C66 = 16.1000

VARIATIONAL BOUNDS OF BULK MODULI	31.8610	31.3909	AVERAGE = 31.6260
VARIATIONAL BOUNDS OF SHEAR MODULI	10.5431	9.7118	AVERAGE = 10.1275
VARIATIONAL BOUNDS OF YOUNGS MODULI	28.4872	26.4116	AVERAGE = 27.4494
VOIGT AND REUSS BOUNDS OF BULK MODULI	32.4111	30.5796	AVERAGE = 31.4954
VOIGT AND REUSS BOUNDS OF SHEAR MODULI	11.8067	8.6064	AVERAGE = 10.2065
VOIGT AND REUSS BOUNDS OF YOUNGS MODULI	31.5848	23.6048	AVERAGE = 27.5948

TIN
TETRAGONAL SYSTEM
INPUT DATA C11 = 82.7400 C12 = 57.8500 C13 = 34.2100 C33 = 103.1000 C44 = 26.9500 C66 = 28.1800

VARIATIONAL BOUNDS OF BULK MODULI	57.8962	57.8943	AVERAGE = 57.8952
VARIATIONAL BOUNDS OF SHEAR MODULI	24.8076	24.3113	AVERAGE = 24.5594
VARIATIONAL BOUNDS OF YOUNGS MODULI	65.1216	63.9785	AVERAGE = 64.5500
VOIGT AND REUSS BOUNDS OF BULK MODULI	57.9022	57.8908	AVERAGE = 57.8965
VOIGT AND REUSS BOUNDS OF SHEAR MODULI	25.9033	22.5772	AVERAGE = 24.2403
VOIGT AND REUSS BOUNDS OF YOUNGS MODULI	67.6256	59.9395	AVERAGE = 63.7825

UNCLASSIFIED

Security Classification

DOCUMENT CONTROL DATA - R & D

(Security classification of title, body of abstract and indexing annotation must be entered when the overall report is classified)

1. ORIGINATING ACTIVITY *(Corporate author)* Naval Research Laboratory Washington, D.C. 20390	2a. REPORT SECURITY CLASSIFICATION Unclassified 2b. GROUP

3. REPORT TITLE

A Computer Program for Determining the Polycrystalline Elastic Moduli from the Single-Crystal Elastic Constants

4. DESCRIPTIVE NOTES *(Type of report and inclusive dates)*

Final report on one phase of a continuing problem.

5. AUTHOR(S) *(First name, middle initial, last name)*

Carl L. Vold

6. REPORT DATE November 30, 1971	7a. TOTAL NO. OF PAGES 20	7b. NO. OF REFS 18
8a. CONTRACT OR GRANT NO. Problem No. M01-23 b. PROJECT NO. RR002-01-46-5430 c. d.	9a. ORIGINATOR'S REPORT NUMBER(S) NRL Report 7333 9b. OTHER REPORT NO(S) *(Any other numbers that may be assigned this report)*	

10. DISTRIBUTION STATEMENT

Approved for public release; distribution unlimited.

11. SUPPLEMENTARY NOTES	12. SPONSORING MILITARY ACTIVITY Dept. of the Navy (Office of Naval Research), Arlington, Va. 22217

13. ABSTRACT

A computer program has been written to evaluate the effective elastic moduli of a quasi-isotropic (random) polycrystalline aggregate in terms of the single-crystal elastic constants. These moduli are calculated using the variational methods developed by Hashin and Shtrikman, and establish stringent limits on the upper and lower bounds for the effective elastic moduli. The program may be applied to crystals of cubic (all point groups), hexagonal (all point groups), trigonal (point groups $\bar{3}$m, 32, and 3m) and tetragonal (point groups 4/mmm, $\bar{4}$2m, 4mm, and 422) symmetries.

DD FORM 1 NOV 65 1473 (PAGE 1)

S/N 0101-807-6801

Security Classification

Security Classification

14. KEY WORDS	LINK A		LINK B		LINK C	
	ROLE	WT	ROLE	WT	ROLE	WT
Polycrystals						
Elastic properties						
Modulus of elasticity						
Computer program						